Università degli Studi di Torino

Corso di Laurea Magistrale in Fisica dei Sistemi Complessi

Tesina di Econofisica

Dynamics of a financial market with heterogeneous agents

Ludovica Antonozzi

Anno Accademico 2018-2019

Abstract

The aim of this work is to reproduce the dynamics of a financial market with heterogeneous agents. Some of them are influenced by the sentiment of their closest colleagues regarding the future evolution of the market, others act in a rational way, while a third group is composed by completely irrational traders.
The model is essentially based on the work of Thomas Lux and Michele Marchesi, but gets some ideas from other works too. To build it we use NetLogo, an open source agent-based programmable modelling environment.

Contents

Chapter 1

Introduction

In this chapter we will describe the theoretical background of our work. In particular, the main features of the Lux and Marchesi model [1] [2] are reported, together with some aspects of the Efficient Market Hypothesis [3] and of the Johansen, Ledoit, Sornette's model [4] [5].

1.1 Agents

In Lux and Marchesi model, the pool of traders is divided into two groups: the 'fundamentalists' and the 'chartists'. In order to make the market more flexible, in our work we added another type of agents called 'noise traders'.

- ***Fundamentalists***
  These agents follow the premise of the Efficient Market Hypothesis in that they expect prices to reflect the whole of information regarding the value of any asset traded in a financial market. The price of the asset is, thus, never above, nor below its intrinsic value. A fundamentalist trading strategy consists of buying (selling) when the actual market price is believed to be below (above) the fundamental value.

- ***Chartists***
  This group is composed of traders that are only boundedly rational. This means that they act taking into account not only the information that comes to the market, but also attempting to identify price trends and patterns (charts), and considering the behaviour of other traders as a source of information, which results in a tendency towards herding behaviour. Moreover, we further distinguish between optimistic and pessimistic individuals in this group: optimists will buy additional assets as a consequence of a positive sentiment regarding the market, that is, they feel that the news are good and offer good prospects of a future increase in the value of their investments, whereas the pessimists have a negative sentiment about the market and will sell part of their actual holdings. Chartists

are modelled following the idea of Johansen, Ledoit and Sornette [4] [5], with some changes. Firstly, here the chartists are not only influenced by their peers, but also by the fundamentalists. Secondly, the propensity to be influenced by the other chartists is no longer constant and equal for everyone, it differs from individual to individual and fluctuates as time evolves. In particular, the propensity to be contagiated increases if the nature of the news are confirmed by a market movement in the same direction (good (bad) news and price increase (decrease)), and decreases if the nature of the news is in the direction contrary to that of the market (good (bad) news and price decrease (increase)). Summing up all of these informations, four terms contribute to the formation of a chartist's sentiment: the market news, the sentiments of his/her closest colleagues, both chartists and fundamentalists, and an own idiosyncratic aspect of the individual which influences its interpretation of both the news and his friends sentiments.

- ***Noise traders***
  This kind of agents are completely irrational: they decide to buy or sell assets without any reasoning, in a perfectly random way.

1.2 Dynamics of the artificial market

The main building blocks of the Lux and Marchesi model are movements of individuals from one group to another, together with the changes of the fundamental value and the price changes resulting from the agents' market operations. The movements between the fundamentalists and the chartists groups are formalized using exponential functions, so that a switch from one group to another occurs with a certain time-varying probability

$$\nu \exp(U(t))\Delta t \tag{1.1}$$

within some small time increment Δt. Here the coefficient ν is a parameter for the frequency of revaluation of opinion or strategy by the agents (possibly assuming different values for different types of switches), and the function $U(t)$ is a forcing term covering those factors that are decisive for the changes of behaviour.
The dynamics of the artificial market are composed of the following elements:

- ***Chartists' changes of opinion***.
  The probabilities for these changes, from a pessimistic to an optimistic mood and vice versa, during a small time increment Δt, are given by $\pi_{+-}\Delta t$ and $\pi_{-+}\Delta t$ and are concretised as follows:

$$\pi_{+-} = \nu_1 \frac{n_c}{N} \exp(U_1) \tag{1.2}$$

$$\pi_{-+} = \nu_1 \frac{n_c}{N} \exp(-U_1) \tag{1.3}$$

$$U_1 = \alpha_1 x + \frac{\alpha_2}{\nu_1} \frac{dp}{dt} \frac{1}{p} \tag{1.4}$$

where π_{+-} (π_{-+}) is the rate from state optimist to state pessimist (from state pessimist to state optimist), N is the total number of agents operating in the artificial market, n_c is the number of chartists, and ν_1, α_1 and α_2 are parameters that measure the frequency of revaluation of opinion, the importance of 'flows' (i.e. the observed behaviour of others) and the importance of charts, respectively. Furthermore the transition probabilities are multiplied by the actual fraction of chartists, because chartists are also allowed to interact with fundamental traders. We can see that the basic influences acting on the chartists' formation of opinion are the majority opinion of their fellow traders, $x = \frac{n_+ - n_-}{n_c}$, and the actual price trend, $\frac{dp}{dt}\frac{1}{p}$. If both components are in harmony, a dominant trend of switches will ensue. For example, observation of price increases together with a prevalence of optimistic individuals would be seen as a strong indication of a continuation of a rising market, and would result in a tendency of formerly pessimistic individuals to convert to the optimistic group. Conflicting signals, of course, would reduce the disposition to follow either the majority opinion or the actual price trend.

- ***Switches between the chartists and the fundamentalists***.
  This time four transition probabilities are needed:

$$\pi_{+f} = \nu_2 \frac{n_+}{N} \exp(U_{2,1}) \tag{1.5}$$

$$\pi_{f+} = \nu_2 \frac{n_f}{N} \exp(-U_{2,1}) \tag{1.6}$$

$$\pi_{-f} = \nu_2 \frac{n_-}{N} \exp(U_{2,2}) \tag{1.7}$$

$$\pi_{f-} = \nu_2 \frac{n_f}{N} \exp(-U_{2,2}) \tag{1.8}$$

where ν_2 is the reaction coefficient for the frequency with which agents reconsider appropriateness of their trading strategy, n_+, n_- and n_f are the number of optimist chartists, of pessimist chartists and of fundamentalists, respectively.
$U_{2,1}$, $U_{2,2}$ are the forcing terms for the transitions regarding optimists and pessimists, respectively:

$$U_{2,1} = \alpha_3 \left\{ \frac{r + \frac{1}{\nu_2}\frac{dp}{dt}}{p} - R - s \left| \frac{p_f - p}{p} \right| \right\} \tag{1.9}$$

$$U_{2,2} = \alpha_3 \left\{ R - \frac{r + \frac{1}{\nu_2}\frac{dp}{dt}}{p} - s \left| \frac{p_f - p}{p} \right| \right\} \tag{1.10}$$

where α_3 is a reaction coefficient for the agents' sensitivity to profit differentials. These equations show that the switches are driven by the difference between the (momentary) profits earned by individuals in both groups. Excess profits (compared to alternative investments) enjoyed by chartists from the optimistic group are composed of nominal dividends (r) and capital gains due to the price change (dp/dt). Dividing by the actual market price p gives the revenue per unit of the asset. Excess returns compared to other investment opportunities are computed by subtracting the average real returns (R) received by the holders of other assets in our economy. Fundamentalists, on the other hand, consider the deviation between price and fundamental value p_f (irrespective of its sign) as the source of arbitrage opportunities. The difference between both strategies is that the gains of chartists are immediately realized whereas those claimed by fundamentalists occur only in the future and depend on the uncertain time for reversal to the fundamental value. To take account of this, fundamentalists' profits are discounted by a factor $s < 1$. The difference between profits of both groups enters the transition probability through the U-function, with switches from the fundamentalists group to the (optimistic) chartists dominating if the profit differential is in favour of the latter group and vice versa. For comparison of profits between pessimistic chartists and fundamentalists, the point of view has to be changed appropriately: because the former rush out of the market in order to avoid losses, their gain is given by the difference between the average profit rate from alternative investments (assumed to be constant) minus the price change (which, when negative, amounts to a capital loss) of the asset they sell. Again, a profit differential in favour of one of the two groups tends to induce changes of behaviour among members of the other group.

- ***Switches between the noise traders and the other two groups***.
  Since noise traders are considered to be completely irrational, their switches will be random, as their decisions. So, we can assume that occasionally one noise trader begins to act like a chartist or like a fundamentalist. On the other hand, we have to consider that fundamentalists and chartists might randomly become crazy and begin to act without any reasoning, entering the noise traders' group.

- ***Price changes***.
  These are endogenous responses of the market to imbalances between demand and supply, excess demand (supply) leading to an increase (decrease) of the prevailing

price. Demand and supply, however, originate from the decisions of our agents: assuming a constant average trading volume of chartists and noise traders, their demand and supply is readily determined by the actual numbers of optimistic and pessimistic individuals. Fundamentalists' sensitivity to the relative deviation of the price from the fundamental value, on the other hand, amounts to an excess demand of this group depending on the difference $p - p_f$. Overall excess demand is the sum of both components.

- ***Changes of fundamental value***.
  These constitute the external driving force which affects the market through the operations of fundamentalist traders. We assume that the relative changes of p_f are Gaussian random variables, i.e. $ln(p_{f,t}) - ln(p_{f,t-1}) = \epsilon_t$ with ϵ_t following a normal distribution with average value zero and time-invariant variance σ_ϵ^2.

Chapter 2

The Model

Our work tries to simulate the model described in the previous section using NetLogo modelling environment. It can be considered a revised version of the Artificial Financial Market Model [6] that can be found in the NetLogo Models Library. The main difference between our model and the Artificial Stock Market Model is the introduction of two new kinds of agents: one that acts on a rational basis, and makes its decisions on the basis of the real value of the asset, the other that is completely irrational and acts randomly. The market is simplified, as it contains just one kind of asset and the operators can trade only one unit per time. Every period, news (good or bad) reach the market, and each operator can choose between selling and buying an asset.

2.1 Interface

2.1.1 The World

The world is composed of 1089 patches, each of them representing one trader (Figure 2.1). They are differently coloured depending on the kind of the agent: yellow for the noise traders, white for the fundamentalists and black for the chartists (only at first, than, when the simulation starts, they divide into optimists (green coloured) and pessimists (red coloured)).

2.1.2 Buttons and Sliders

In the interface there are several buttons and sliders (Figure 2.2).
The buttons are three:

- ***Setup button***: it creates the traders and initializes all the variables.
- ***Go button***: it is a forever button. Once it is pressed, the simulation starts to run over and over, until it is pressed again.



Figure 2.1: On the left a screen shot of the initial world, on the right a screen shot of the world while the simulation is running

- ***Step once button***: it is a once button. It is like the Go button except that it advances the model only by one time step. It can be used to watch the progress of the model more closely.

The sliders let the user manage the parameters of the simulation:

- ***Fundamentalists slider***: it controls the average percentage of fundamentalists.
- ***Noise Traders slider***: it controls the average percentage of noise traders.
- ***Max-sensitivity-to-news slider***: it controls the chartists' maximum sensitivity to the news.
- ***News-average-value slider***: it controls the average value of the normal distribution through which the news meaning (good or bad) is formalized.
- ***Max-base-propensity-to-chartist-contagion slider***: it controls the maximum base propensity of chartists to be contagiated by their chartist neighbours.
- ***Max-propensity-to-fundamentalist-contagion slider***: it controls the maximum propensity of chartists to be contagiated by their fundamentalist neighbours.
- ***Min-number-of-different-neighbours-needed-to-change-from-f-to-c slider***: it controls the minimum number of fundamentalist neighbours needed to make a chartist become a fundamentalist.
- ***Min-number-of-different-neighbours-needed-to-change-from-c-to-f slider***: it controls the minimum number of chartist neighbours needed to make a fundamentalist become a chartist.
- ***Sigma slider***: it controls the variance of the normal distribution used to formalize the chartists' personal interpretation of the news.



Figure 2.2: Buttons and Sliders

- ***Craziness-probability slider***: it controls the probability that a chartist or a fundamentalist becomes a noise trader.
- ***Switch-probability-from-f-to-c slider***: it controls the probability that a fundamentalist becomes a chartist.
- ***Switch-probability-from-c-to-f slider***: it controls the probability that a chartist becomes a fundamentalist.
- ***Switch-probability-from-nt-to-c slider***: it controls the probability that a noise trader becomes a chartist.
- ***Switch-probability-from-nt-to-f slider***: it controls the probability that a noise trader becomes a fundamentalist.

2.1.3 Plots and Monitors

In the interface there are four plots and some monitors (Figure 2.3) .

- ***Returns plot and monitor***: the plot displays the returns of the market in function of time while the monitor shows its exact value at each time step.
- ***Volatility plot and monitor***: the plot displays the volatility, namely the absolute value of the returns, in function of time while the monitor shows its exact value at each time step.



Figure 2.3: Plots and Monitors

- ***Log-price and Fundamental Value plot and monitors***: the plot displays the trend of both the logarithm of the price and the fundamental value in function of time while the monitors show their exact value at each time step.
- ***Number of Chartists, Fundamentalists and Noise Traders plot and monitors***: the plot displays the number of optimist chartists, pessimist chartists, fundamentalists and noise traders in function of time while the monitors show their exact value at each time step.

2.2 The Code

In this section the most important parts of the code are explained.

2.2.1 Setup

The setup command clears the environment, initializes the parameters and creates the various types of traders, each one with its own colour and variables.

2.2.2 Go

The go button permits to start the simulation. It is composed of various actions that are to be performed during the simulation.

```
to setup
  ca
  set fundamental-value 0
  set log-price 1
  ; fundamentalisits creation
  ask patches
  [ifelse (random-float 100) < fundamentalists
    [set pcolor white]
    [ifelse (random-float 100) < noise-traders
       [set pcolor yellow]
       [set pcolor black]]]
  ask patches with [pcolor = white]
    [set number-of-assets 1
     set fundamentalists-counter 1
     set number-of-different-neighbors-needed-to-change-from-f-to-c min-number-of-different-neighbors-needed-to-change-from-f-to-c
    ]
  ; noise-traders creation
  ask patches with [pcolor = yellow]
    [set number-of-assets 1
     set noise-traders-counter 1]
   ; chartists creation. Notice that each chartist has different values of the parameters.
   ask patches with [pcolor = black]
    [set number-of-assets 1
     set variance-of-chartists-personal-interpretation-of-news sigma + random-float 0.1
     set sensitivity-to-news (random-float max-sensitivity-to-news)
     set base-propensity-to-chartists-contagion (random-float max-base-propensity-to-chartists-contagion)
     set propensity-to-chartists-contagion base-propensity-to-chartists-contagion
     set propensity-to-fundamentalists-contagion (random-float max-propensity-to-fundamentalists-contagion)
     set number-of-different-neighbors-needed-to-change-from-c-to-f min-number-of-different-neighbors-needed-to-change-from-c-to-f
     set chartists-counter 1
    ]
end
```

```
to go
  news-arrival
  chartists-decision
  fundamentalists-decision
  noise-traders-decision
  return-price-computation
  update-propensity-to-chartists-contagion
  categories-switch
  do-plot
end
```

News arrival

At each time step news arrive to the market. It is assumed that this information is a random quantity normally distributed with standard deviation 1 and average value controlled by the user through the slider 'news-average-value'. A qualitative meaning is given to the information, this meaning is taken in binary terms as either being good (+1) or bad (-1). The news qualitative meaning determines the fundamental value of shares. If the news are good it increases, if the news are bad it decreases.

Chartists' decision

Chartists are boundedly rational. Their decision to buy or sell is not only based on the news qualitative meaning, but it depends on other factors, too. In particular, the chartists' sentiment depends on four terms: the influence of chartists neighbours, the

```
to news-arrival
  ifelse (random-normal news-mean 1) > 0
  [set news-qualitative-meaning 1
   set fundamental-value fundamental-value + random-float 1.00]
  [set news-qualitative-meaning -1
   set fundamental-value fundamental-value - random-float 1.00]
end


to chartists-decision
  ask patches with [pcolor != white and pcolor != yellow]
  [ifelse ((propensity-to-chartists-contagion * sum [chartist-sentiment] of neighbors
    with [pcolor != white and pcolor != yellow] + propensity-to-fundamentalists-contagion *
    sum [fundamentalist-sentiment] of neighbors with [pcolor = white] +
    sensitivity-to-news * news-qualitative-meaning +
    random-normal 0 variance-of-chartists-personal-interpretation-of-news) > 0)
    [set chartist-sentiment 1
     set number-of-assets number-of-assets + 1
     set optimists-counter 1
     set pessimists-counter 0]
    [set chartist-sentiment -1
     set number-of-assets number-of-assets - 1
     set pessimists-counter 1
     set optimists-counter 0]]
  ask patches with [pcolor != white and pcolor != yellow]
  [if chartist-sentiment = 1
    [set pcolor green]
   if chartist-sentiment = -1
    [set pcolor red]]
end
```

influence of fundamentalists neighbours, the news qualitative meaning and a random term that accounts for each individual's own interpretation of the news (we take it to be normally distributed around zero and with standard deviation to be controlled by the user through the slider 'sigma'). Note that each of the four elements which determine the chartists' sentiment differ from trader to trader, except the news qualitative meaning which is a common component. If the sum of these terms is positive, than the chartist is optimistic and buys an asset. If the sum is negative the chartist is pessimistic and sells an asset. We assume that each trader can buy or sell only one asset per period.

Fundamentalists' decision

Fundamentalists are absolutely rational. They decide to buy or sell assets depending only on the fundamental value of the asset. If the fundamental value of the asset is below its market price they sell it, if the fundamental value is above the market price they buy the asset. We assume that each trader can buy or sell only one asset per period.

Noise traders' decision

The noise traders are completely irrational. They decide to buy or sell assets randomly.

```
to fundamentalists-decision
  ask patches with [pcolor = white]
  [ifelse (fundamental-value > log-price)
    [set fundamentalist-sentiment 1
     set number-of-assets number-of-assets + 1]
    [set fundamentalist-sentiment -1
     set number-of-assets number-of-assets - 1]]
end

to noise-traders-decision
  ask patches with [pcolor = yellow]
  [ifelse (random-float 1.0 > 0.5)
    [set noise-traders-sentiment 1
     set number-of-assets number-of-assets + 1]
    [set noise-traders-sentiment -1
     set number-of-assets number-of-assets - 1]]
end
```

Returns and price computation

We take the logarithm of the price to be the sum of the previous period price plus the sum of the traders sentiment divided by the number of traders. The difference between the present log-price and the previous log-price, that is obviously the sum of the traders sentiment divided by the number of traders, is defined as returns. Moreover we define the volatility as the absolute value of the returns. It is worth to note that the price formation rule is quite bizarre, since it makes the price variation rate (that is called return in the code) increasing or decreasing too rapidly with the global sentiment of the market (namely the sum of the sentiments of all the traders, that is the numerator of the return). For example, as we can see in the right plot of 2.4, when the number of optimists and pessimists differs by one unit, the price variation rate is 0.00092; when it differs by two units, the price variation rate becomes 0.00184, two times higher than before; when it differs by three units, the price variation rate becomes 0.00276, three times higher; and so on.

Propensity to chartists' contagion update

We assume that chartists have a base propensity to be contagiated by the sentiment of others, and that, if a good (bad) news is confirmed by a market movement in the same direction, then the individual's propensity to contagion is set equal to his base propen-

```
to return-price-computation
  set return-denominator count patches
  set return-numerator1 sum [chartist-sentiment] of patches with [pcolor != white and pcolor != yellow ]
  set return-numerator2 sum [fundamentalist-sentiment] of patches with [pcolor = white]
  set return-numerator3 sum [noise-traders-sentiment] of patches with [pcolor = yellow]
  set return-numerator return-numerator1 + return-numerator2 + return-numerator3
  set return return-numerator / return-denominator
  set log-price log-price + return
  set volatility abs return
end
```



Figure 2.4: Plot of the price variation rate versus the total sentiment of the market's traders. On the left the complete plot, on the right a zoom of the plot for values of the total sentiment within a [-10,10] range.

```
to update-propensity-to-chartists-contagion
  ask patches with [pcolor != white and pcolor != yellow]
  [if (return > 0) and (news-qualitative-meaning > 0)
  [set propensity-to-chartists-contagion  base-propensity-to-chartists-contagion + return]
  if (return > 0) and (news-qualitative-meaning < 0)
  [set propensity-to-chartists-contagion  base-propensity-to-chartists-contagion - return]
  if (return < 0) and (news-qualitative-meaning < 0)
  [set propensity-to-chartists-contagion  base-propensity-to-chartists-contagion - return]
  if (return < 0) and (news-qualitative-meaning > 0)
  [set propensity-to-chartists-contagion  base-propensity-to-chartists-contagion + return]]
end
```

sity plus (less) an amount equal to the value of the returns, otherwise the propensity to contagion is set equal to the base propensity less (plus) an amount equal to the value of the returns.

Switch of categories

At the end of every time step each patch has the possibility to change the type. The switches from chartists to fundamentalists and vice versa happen depending on the number of a trader's neighbours of different type. If the number of neighbours of different type is above a certain threshold, then the trader has a certain chance to change, if it is not, then the trader will not change his type. The switches from chartists or fundamentalists to noise traders and vice versa happen randomly, according to some probabilities setted by the user through the sliders 'craziness-probability' (for the switch from fundamentalists or chartists to noise traders), 'switch-probability-from-nt-to-c' (for the switch from noise traders to chartists) and 'switch-probability-from-nt-to-f' (for the switch from noise traders to fundamentalists).

```
to categories-switch
  ; switch from chartist to noise-trader
  ask patches with [pcolor != white and pcolor != yellow]
  [if (random-float 100 < craziness-probability)
    [set pcolor yellow
     set fundamentalists-counter 0
     set chartists-counter 0
     set pessimists-counter 0
     set optimists-counter 0
     set noise-traders-counter 1]]
  ; switch from chartist to fundamentalist
  ask patches with [pcolor != white and pcolor != yellow]
  [set number-of-fundamentalists sum [fundamentalists-counter] of neighbors with [pcolor = white]]
  ask patches with [pcolor != white and pcolor != yellow]
  [if (number-of-fundamentalists) > number-of-different-neighbors-needed-to-change-from-c-to-f
    [ask patches with [pcolor != white and pcolor != yellow]
      [if (random-float 100) < switch-probability-from-c-to-f
        [set pcolor white
         set number-of-different-neighbors-needed-to-change-from-f-to-c min-number-of-different-neighbors-needed-to-change-from-f-to-c
         set fundamentalists-counter 1
         set chartists-counter 0
         set pessimists-counter 0
         set optimists-counter 0
         set noise-traders-counter 0]]]
  ; switch from fundamentalist to noise-trader
  ask patches with [pcolor = white]
   [if (random-float 100 < craziness-probability)
    [set pcolor yellow
     set fundamentalists-counter 0
     set chartists-counter 0
     set pessimists-counter 0
     set optimists-counter 0
     set noise-traders-counter 1]]
  ; switch from fundamentalist to chartist
  ask patches with [pcolor = white]
  [set number-of-optimists sum [optimists-counter] of neighbors with [pcolor != white and pcolor != yellow]
   set number-of-pessimists sum [pessimists-counter] of neighbors with [pcolor != white and pcolor != yellow]]
  ask patches with [pcolor = white]
  [if (number-of-pessimists) > number-of-different-neighbors-needed-to-change-from-f-to-c
  [fundamentalist-to-pessimist-switch]]
  ask patches with [pcolor = white]
  [if (number-of-optimists) > number-of-different-neighbors-needed-to-change-from-f-to-c
  [fundamentalist-to-optimist-switch]]
  ; switch from noise-trader to chartist
  ask patches with [pcolor = yellow]
  [if (random-float 100 < switch-probability-from-nt-to-c)
      [ifelse (random-float 100 < 50)
       [set pcolor red
        set variance-of-chartists-personal-interpretation-of-news (sigma + random-float 0.1)
        set sensitivity-to-news (random-float max-sensitivity-to-news)
        set base-propensity-to-chartists-contagion (random-float max-base-propensity-to-chartists-contagion)
        set propensity-to-chartists-contagion base-propensity-to-chartists-contagion
        set number-of-different-neighbors-needed-to-change-from-c-to-f min-number-of-different-neighbors-needed-to-change-from-c-to-f
        set chartists-counter 1
        set pessimists-counter 1
        set fundamentalists-counter 0
        set noise-traders-counter 0]
       [set pcolor green
        set variance-of-chartists-personal-interpretation-of-news (sigma + random-float 0.1)
        set sensitivity-to-news (random-float max-sensitivity-to-news)
        set base-propensity-to-chartists-contagion (random-float max-base-propensity-to-chartists-contagion)
        set propensity-to-chartists-contagion base-propensity-to-chartists-contagion
        set number-of-different-neighbors-needed-to-change-from-c-to-f min-number-of-different-neighbors-needed-to-change-from-c-to-f
        set chartists-counter 1
        set optimists-counter 1
        set fundamentalists-counter 0
        set noise-traders-counter 0]]]
```

```
; switch from noise-trader to fundamentalist
  ask patches with [pcolor = yellow]
  [if (random-float 100 < switch-probability-from-nt-to-f)
    [set pcolor white
     set number-of-different-neighbors-needed-to-change-from-f-to-c min-number-of-different-neighbors-needed-to-change-from-f-to-c
     set fundamentalists-counter 1
     set chartists-counter 0
     set pessimists-counter 0
     set optimists-counter 0
     set noise-traders-counter 0]]
  set number-of-fundamentalists sum [fundamentalists-counter] of patches with [pcolor = white]
  set number-of-pessimists sum [pessimists-counter] of patches with [pcolor != white and pcolor != yellow]
  set number-of-optimists sum [optimists-counter] of patches with [pcolor != white and pcolor != yellow]
  set number-of-noise-traders sum [noise-traders-counter] of patches with [pcolor = yellow]
end
  to fundamentalist-to-pessimist-switch
    ask patches with [pcolor = white]
    [if (random-float 100) < switch-probability-from-f-to-c
    [set pcolor red
    set variance-of-chartists-personal-interpretation-of-news (sigma + random-float 0.1)
    set sensitivity-to-news (random-float max-sensitivity-to-news)
    set base-propensity-to-chartists-contagion (random-float max-base-propensity-to-chartists-contagion)
    set propensity-to-chartists-contagion base-propensity-to-chartists-contagion
    set number-of-different-neighbors-needed-to-change-from-c-to-f min-number-of-different-neighbors-needed-to-change-from-c-to-f
    set chartists-counter 1
    set pessimists-counter 1
    set fundamentalists-counter 0
      set noise-traders-counter 0]]
   end
   to fundamentalist-to-optimist-switch
    ask patches with [pcolor = white]
    [if (random-float 100) < switch-probability-from-f-to-c
    [set pcolor green
    set variance-of-chartists-personal-interpretation-of-news (sigma + random-float 0.1)
    set sensitivity-to-news (random-float max-sensitivity-to-news)
    set base-propensity-to-chartists-contagion (random-float max-base-propensity-to-chartists-contagion)
    set propensity-to-chartists-contagion base-propensity-to-chartists-contagion
    set number-of-different-neighbors-needed-to-change-from-c-to-f min-number-of-different-neighbors-needed-to-change-from-c-to-f
    set chartists-counter 1
    set optimists-counter 1
    set fundamentalists-counter 0
    set noise-traders-counter 0]]
   end
```

Chapter 3

Results

Running the simulation with traders almost equally divided into the three categories and the other parameters fixed to medium values leads to a very complex dynamics. Traders are well mixed, and segregation between different groups never emerges. This is due to the existence of fundamentalist and noise traders operators, who act regardless of their neighbours. Prices more or less follow the fundamental value and they both show very frequent crashes and bubbles, normally due to an unusual dominance of one mood over the other in the chartists (see Figure 3.1). This is evident in the returns and volatility plots too (Figure 3.2). Large deviations of the price from the fundamental value are avoided through endogenous mechanisms: they would be seen as profit opportunities or losses risks by fundamentalists whose operations then tend to stabilize the market again. It is interesting to investigate what happens when there is a predominance of one type of traders over the others:

- When most of the agents are noise traders, the price and the fundamental value show the same trend, even if the price varies much slower (Figure 3.3). This is because the fundamental value exactly reflects the news and increase or decrease each time one of them arrives to the market. Instead, the price changes are based on the general sentiment of the market, and in this case it is very unlikely to have a prevalence of a positive or negative mood among traders, since their decisions are random. As a consequence, the values of both the returns and the volatility are always very small (see Figure 3.4). Anyway, both the movements of price and fundamental value can be described by a random walk model, where changes have the same distribution and are independent of each other, therefore, the past trends cannot be used to predict future movements.

- When most of the agents are chartists the price don't exactly follow the trend of the fundamental value. In fact, the price formation rule takes into account the global sentiment of the market, and if the traders are mainly chartists, their sentiment doesn't reflect the real value of the assets but it is influenced by other factors. It is worth noting that even if chartists make predictions on the future evolution of the prices without any reasonable or scientific basis, their predictions



Figure 3.1: Traders equally distributed in the three groups.
On the top, the plot of the number of fundamentalists, chartists and noise traders. On the bottom, the plot of the log-price and fundamental value trends.



Figure 3.2: Returns and volatility plot when the traders are almost equally divided into the three cathegories



Figure 3.3: Prevalence of noise traders.
On the top, the plot of the number of fundamentalists, chartists and noise traders. On the bottom, the plot of the log-price and fundamental value trends.



Figure 3.4: Returns and volatility plot when most of the agents are noise traders.

come true. We can see in Figure 3.5 that when there is a prevalence of pessimists the price actually goes down while when the most are optimists the price raises. This happens not because chartists' methods and previsions are good, but because here chartists do make the market. They are the majority of the traders, and they all follow their own predictions, so these will inevitably come true. For example, if everyone believes that prices will raise, traders will buy assets, and prices will actually rise; conversely, if everyone believes that prices will go down, traders will sell assets and prices will actually go down. The prevalence of one mood over the other is evident also looking at the returns and volatility plots (Figure 3.6). When the number of optimists and pessimists is almost the same, there are mild fluctuations in the prices, the returns are small and so is the volatility. On the contrary, when there is a prevalence of optimists (or pessimists) the returns become larger (or smaller) and create a pick, together with the volatility. We can say that the smaller the herding, the smaller are the returns, be they positive or negative, on the other hand, the higher the herding, the larger are the returns. The maximum herding would be everyone sharing the same sentiment, in this case the absolute returns would be one. Another interesting aspect to analyse is the segregation that emerges between optimists and pessimists. It seem to be positively correlated to the propensity to chartist's contagion slider: an increment in its value leading to a more sharp polarization. Such a situation doesn't emerge when the number of fundamentalists and noise traders in the market is higher, this is because these two kinds of operators act regardless of their neighbours.

- When most of the agents are fundamentalists the price exactly reflects the fundamental value. We analysed three situations, modifying the average value of the news meaning. When it is set to zero, the price and the fundamental value show some oscillations but there isn't an actual rising or falling trend (Figure 3.7). Conversely, when the average value of the news meaning is positive, there is a noticeable rising trend (see Figure 3.8). This happens because the news will be most of the times positive, and so the fundamental value will increase by definition, fundamentalists will have a positive sentiment and will buy assets. Furthermore, since fundamentalists are the most of the traders, they will have a big influence on the global sentiment of the market, guiding the price trend to rise. The opposite happens when the average value of the news meaning is negative: the news will be most of the times negative, the fundamental value will decrease, fundamentalists will have a negative sentiment and will sell assets and the price will go down (see Figure 3.9). Generally speaking, we can say that a rational behaviour can enhance the chances of abnormal movements of the prices as the rational agents move in the same direction at the same time. This is evident also in the returns and volatility plots (Figure 3.10). Since the fundamentalists always act in the same way, the returns are always maximum: say +1 or -1, and so is the volatility, which is always +1.



Figure 3.5: Prevalence of chartists.
On the top, the plot of the number of fundamentalists, chartists and noise traders. On the bottom, the plot of the log-price and fundamental value trends.



Figure 3.6: Returns and volatility plot when most of the traders are chartists.



Figure 3.7: Prevalence of fundamentalists with news meaning average value set to zero.
On the top, the plot of the number of fundamentalists, chartists and noise traders. On the bottom, the plot of the log-price and fundamental value trends.



Figure 3.8: Prevalence of fundamentalists with positive news meaning average value.
On the top, the plot of the number of fundamentalists, chartists and noise traders. On the bottom, the plot of the log-price and fundamental value trends.



Figure 3.9: Prevalence of fundamentalists with negative news meaning average value.
On the top, the plot of the number of fundamentalists, chartists and noise traders. On the bottom, the plot of the log-price and fundamental value trends.



Figure 3.10: Returns and volatility plot when most of the traders are fundamentalists.

Bibliography

[1] Lux T., Marchesi M.: Scaling and criticality in a stochastic multi-agent model of a financial market; Nature 397, 498 (1999).

[2] Samanidou E., Zschischang E., Stauffer D., Lux T.: Agent-based models of financial markets; Rep. Prog. Phys., 70, 438–442 (2007)

[3] Burton G. M.: The efficient market hypothesis and its critics; Journal of Economic Perspectives 17, 59:82 (2003).

[4] Johansen A., Ledoit O., Sornette D.: Crashes as critical points; Int. J. Theor. Appl. Fin. 3, 219:255 (2002).

[5] Sornette, D.: Critical market crashes; Phys. Rep. 378, 1:98 (2003).

[6] Gonçalves C. P.: Artificial Financial Market Model; `http://ccl.northwestern.edu/netlogo/models/community/Artificial%20Financial%20Market` (2003)